2 - 2

0

Parrondo's Paradox 3

Nicholas Wheeler
3 October 2013

Introduction

In Parts 1 & 2 of this series of notebooks I described two distinct computational illustrations of Parrondo's Paradox. In Part 1—drawing my inspiration from recent study (*Mathematica* notebook dated 22 September 2013) of the "Stanley's digital ratchet"

Youngki **L**ee, Andrew **A**llison, Derek **A**bbott & H. Eugene **S**tanley, "Minimal Brownian Ratchet: An Exactly Solvable Model," PRL **91**, 220601 (2003)

—I phrased the issue as it relates to a class of random walks on a 3-vertex graph; it was, indeed, LAAS's reference to

G. P. Harmer & D Abbot, "Parrondo's Paradox," Statistical Science 14, 206-213 (1999)

that brought Parrondo's paradox to my attention. In Part 2 I look to the statistical consequences of playing many-flip Parrondo games many times (many random walks on an infinite lattice). The two approaches to the problem involve quite different analytical methods, have quite a different feel, and yield results that are conclusive in the former instance, only statistically suggestive in the latter, but in general agreement.

But in neither of those notebooks did I attempt to indicate **WHY** the Parrondo game leads to its paradoxical results. That is the issue with which I attempt to deal here.

With the equivalence of the two approaches established, the first—from an analytical viewpoint—has much to recommend it. It is finitistic, presents the Parrondo phenomenon as a property of a certain class of Markov processes that act on 3-dimensional stochastic state vectors, and avoids all reference to the statistics of many long games. In this approach to the WHY question I adopt the first (finite Markovian) approach…but may later attempt to expose the foundations of the statistical approach.

Setting things up

Random walks on the following graph (NOTE: The commands that generated the graphics used in this notebook are stored in a separate notebook: Parnondo's Game Graphics)



are generated by Markov matrices of the form

$$\begin{pmatrix} M_{1\leftarrow 1} & M_{1\leftarrow 2} & M_{1\leftarrow 3} \\ M_{2\leftarrow 1} & M_{2\leftarrow 2} & M_{2\leftarrow 3} \\ M_{3\leftarrow 1} & M_{3\leftarrow 2} & M_{3\leftarrow 3} \end{pmatrix}$$

where each column is stochastic (non-negative elements that sum to unity). We will be concerned with walks in which the next step (to one nearest neighbor or the other) is determined by flipping the coin that lives at the presently occupied vertex. Coins don't provide a "stand in place" option, so the diagonal elements of the matrix vanish. The resulting matrix is a 3-parameter object, of the form

$$\mathbb{M} = \begin{pmatrix} 0 & 1-y & z \\ x & 0 & 1-z \\ 1-x & y & 0 \end{pmatrix};$$

where {x, y, z} range on the unit interval.

REMARK: Diagonal elements would appear if (for example) we associated with each vertex a die, with {0, H, T} inscribed on opposite faces. But ther additional three parameters would greatly complicate the analysis. It would, in particular, deprive us of some valuable graphic resources.

We have

```
Eigenvalues[M]
```

$$\left\{1, \frac{1}{2}\left(-1-\sqrt{-3+4x+4y-4xy+4z-4xz-4yz}\right), \frac{1}{2}\left(-1+\sqrt{-3+4x+4y-4xy+4z-4xz-4yz}\right)\right\}$$

Note the occurance here (and below) of the symmetric polynomials

$x+y+z$ and $xy+yz+zx$

```
Simplify[-3 + 4 x + 4 y - 4 x y + 4 z - 4 x z - 4 y z == 4 (x + y + z) - 4 (x y + y z + z x) - 3]
```

True

The leading eigenvalue is unity (the others are typically complex, with amplitudes < 1) and the associated eigenvector is

```
Eigenvectors[M][[1]]
```

$$\left\{-\frac{-1+y-yz}{1-x+xy}, -\frac{-1+z-xz}{1-x+xy}, 1\right\}$$

Multiply by the shared denominator and divide by the sum

$$\text{Simplify}\left[(-1+x-xy)\left\{-\frac{-1+y-yz}{1-x+xy}, -\frac{-1+z-xz}{1-x+xy}, 1\right\}\right]$$

$$\{-1+y-yz, -1+z-xz, -1+x-xy\}$$

```
μ = Total[%]
```

$$-3+x+y-xy+z-xz-yz$$

(not that μ is again assembled from symmetric polynomials) to obtain the stochastic eigenvector

$$\text{m} = \text{Transpose}\left[\left\{\frac{\{-1+y-yz, -1+z-xz, -1+x-xy\}}{\mu}\right\}\right];$$

```
m // MatrixForm
```

$$\begin{pmatrix} \frac{-1+y-yz}{-3+x+y-xy+z-xz-yz} \\ \frac{-1+z-xz}{-3+x+y-xy+z-xz-yz} \\ \frac{-1+x-xy}{-3+x+y-xy+z-xz-yz} \end{pmatrix}$$

```
Total[Transpose[m][[1]]] // Simplify
Simplify[M.m == m]
```

```
1
```

```
True
```

The eigenvector **m** acquires its special importance from the circumstance that for all initial stochastic vectors

$$P = \begin{pmatrix} p_1 \\ p_2 \\ p_3 \end{pmatrix}$$

one has

$$\text{Limit}\left[\text{MatrixPower}[\mathbb{M}, n].P,\ n \to \infty\right] == m$$

so **m** describes the **asympototic steady state**.

The current vector

The action of a Markov matrix can be described in terms of the "probability currents" that it sets up, the basic concept being

$$\text{Current}_{j\leftarrow i} = M_{j\leftarrow i}\, p_i$$

$$\sum_{\text{neighbors } j \text{ of } i} \text{Current}_{j\leftarrow i} = \text{single iteration decrease of } p_i$$

In the asymptotic steady state there is "conservation of probability" at every vertex; one has

$$\sum_{\text{neighbors } j \text{ of } i} \left(\text{Current}_{j\leftarrow i} - \text{Current}_{i\leftarrow j}\right) = 0$$

and an analog of Kirchhoff's node law comes naturally into play. The "Kirchhoff matrix" emerges as a kind of "decorated adjacency matrix." The currents $\text{Current}_{j\leftarrow i}$ might, for general graphs, be displayed as elements of an antisymmetric matrix, but for cyclic graphs—such as the grapn of present interest



—they are more efficiently displayed as elements of a current vector, the elements of which acquire their sign from the arbitrary convention that "clockwise means positive". Relative to that convention, we can define "edge currents": thus

$$\text{EdgeCurrent}_{12} = \text{Current}_{2\leftarrow 1} - \text{Current}_{1\leftarrow 2}$$

etc. On the 3-graph it is convenient to give edges the name of the opposite vertex, and to display the edge currents as elements of a vector:

$$\begin{pmatrix} j_1 \\ j_2 \\ j_3 \end{pmatrix} = \begin{pmatrix} M_{3\leftarrow 2}\, p_2 - M_{2\leftarrow 3}\, p_3 \\ M_{1\leftarrow 3}\, p_3 - M_{3\leftarrow 1}\, p_1 \\ M_{2\leftarrow 1}\, p_1 - M_{1\leftarrow 2}\, p_2 \end{pmatrix}$$

In the asymptotic steady state we have (at vertex #1, and similarly at other vertices)

$$\text{in}_1 - \text{out}_1 = M_{1\leftarrow 2}\, p_2 + M_{1\leftarrow 3}\, p_3 - M_{2\leftarrow 1}\, p_1 - M_{3\leftarrow 1}\, p_1 = 0$$

which by rearrangement becomes

$$M_{2\leftarrow 1}\, p_1 - M_{1\leftarrow 2}\, p_2 = M_{1\leftarrow 3}\, p_3 - M_{3\leftarrow 1}\, p_1$$

or

$$j_2 = j_3$$

We conclude that in the steady state (1) the probabilities at all vertices become constant, and (2) all edge currents become equal:

$$j_1 = j_2 = j_3 \equiv J$$

Interestingly, **the asymptotic constancy of the probabilities does not imply J = 0**: in the asymptotic state probability typically continues to "circulate," with depletion/replenishment in perpetual balance at every vertex (where the conserved probabilities are typically unequal to each other). The situation resembles perpetual current flow in a superconductor. One is reminded of M. C. Escher's "Waterfall":





■ Parrondo Currents

For the most general "no stand in place" Markov process on a connected 3-graph we have

$$\mathbb{M} = \begin{pmatrix} 0 & 1-y & z \\ x & 0 & 1-z \\ 1-x & y & 0 \end{pmatrix};$$

The asymptotic stochastic eigenvector is, as was already established,

$$P = \begin{pmatrix} \frac{-1+y-y\,z}{-3+x+y-x\,y+z-x\,z-y\,z} \\ \frac{-1+z-x\,z}{-3+x+y-x\,y+z-x\,z-y\,z} \\ \frac{-1+x-x\,y}{-3+x+y-x\,y+z-x\,z-y\,z} \end{pmatrix};$$

```
Total[Transpose[P]⟦1⟧] // Simplify
Simplify[𝕄.P == P]
```

```
1
```

```
True
```

The current vector (in the asymptotic limit) is

```
AsymptoticCurrentVector = Simplify[( M[[3]][[2]] P[[2]][[1]] - M[[2]][[3]] P[[3]][[1]]
                                     M[[1]][[3]] P[[3]][[1]] - M[[3]][[1]] P[[1]][[1]]
                                     M[[2]][[1]] P[[1]][[1]] - M[[1]][[2]] P[[2]][[1]] )]
```

$$\left\{\left\{\frac{-1+y+z-y\,z+x\,(1-z+y\,(-1+2\,z))}{3+y\,(-1+z)-z+x\,(-1+y+z)}\right\},\left\{\frac{-1+y+z-y\,z+x\,(1-z+y\,(-1+2\,z))}{3+y\,(-1+z)-z+x\,(-1+y+z)}\right\},\left\{\frac{-1+y+z-y\,z+x\,(1-z+y\,(-1+2\,z))}{3+y\,(-1+z)-z+x\,(-1+y+z)}\right\}\right\}$$

in which all elements are seen to be—as anticipated—identical. We give them the name

$$J = \frac{-1+y+z-y\,z+x\,(1-z+y\,(-1+2\,z))}{3+y\,(-1+z)-z+x\,(-1+y+z)};$$

The equation J = 0 defines a null surface which bisects the unit cube within which the parameters {x, y, z} range:

```
GeneralNullSurface = ContourPlot3D[J == 0, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}];
Gauge = Graphics3D[{
    {Thick, Line[{{0, 0, 0}, {1, 1, 1}}]},
    {Yellow, Sphere[{0.5, 0.5, 0.5}, {0.02}]},
    {Red, Sphere[{0.25, 0.25, 0.25}, {0.02}]},
    {Blue, Sphere[{0.75, 0.75, 0.75}, {0.02}]}}];
Show[{GeneralNullSurface, Gauge}]
```



From

```
J /. {x → 0, y → 0, z → 0}
J /. {x → 1, y → 1, z → 1}
```

$-\frac{1}{3}$

$\frac{1}{3}$

we see that the red sphere identifies the region in which J < 0, the blue sphere identifies the region in which J > 0. The obvious symmetry of the null surface reflects the fact that J is assembled from symmetric polynomials

1

$x + y + z$

$x y + y z + z x$

$x y z$

The limiting J-value $\frac{1}{3}$ can be understood as a reflection of equidistribution (probability the same at each of the three vertices) and a coin loaded so that it lands heads every time (else tails every time, which flips the sign).

- **The A coin**

Parrondo's A coin lands heads with probability x, ∴ atails with probability 1−x. As a walk on the 3-graph it is modeled by

```
𝔸 = 𝕄 /. {y → x, z → x};
𝔸 // MatrixForm
```

$$\begin{pmatrix} 0 & 1-x & x \\ x & 0 & 1-x \\ 1-x & x & 0 \end{pmatrix}$$

The associated asymptotic current is

```
JA = Simplify[J /. {y → x, z → x}]
```

$\frac{1}{3}(-1+2x)$

One has JA = 0 when the coin is fair: $x = \frac{1}{2}$. The maximal/minimal values of J are $\pm\frac{1}{3}$.

- **The B coins**

Coin B_1—used when the money in the pot **is not a multiple of 3**—lands heads with probability y.
Coin B_2—used when the money in the pot **is a multiple of 3**—lands heads with probability z. The pair (+ their usage rule) are modeled by

```
𝔹 = 𝕄 /. {x → y};
𝔹 // MatrixForm
```

$$\begin{pmatrix} 0 & 1-y & z \\ y & 0 & 1-z \\ 1-y & y & 0 \end{pmatrix}$$

The steady stochastic state has become

```
PB = Simplify[P /. x → y];
PB // MatrixForm
```

$$\begin{pmatrix} \frac{1+y\,(-1+z)}{3+y^2+2\,y\,(-1+z)-z} \\ \frac{1+(-1+y)\,z}{3+y^2+2\,y\,(-1+z)-z} \\ \frac{1-y+y^2}{3+y^2+2\,y\,(-1+z)-z} \end{pmatrix}$$

which possesses the anticipated properties:

```
Total[Transpose[PB][[1]]] // Simplify
Simplify[𝔹.PB == PB]
```

1

True

The associated asymptotic current is

```
JB = Simplify[J /. x → y]
```

$$\frac{-1-2\,y\,(-1+z)+z+y^2\,(-1+2\,z)}{3+y^2+2\,y\,(-1+z)-z}$$

From

```
JB /. {y → 0, z → 0}
JB /. {y → 1, z → 1}
```

$-\frac{1}{3}$

$\frac{1}{3}$

we see that JB < 0 below the following null curve (marked by a red disk), JB > 0 above the curve (blue disk):

```
NullCurveB = ContourPlot[JB == 0, {y, 0, 1}, {z, 0, 1}];
GaugeB = Graphics[{
    Line[{{0, 0}, {1, 1}}],
    {Red, Disk[{0.25, 0.25}, 0.02]},
    {Yellow, Disk[{0.50, 0.50}, 0.02]},
    {Blue, Disk[{0.75, 0.75}, 0.02]}}];
Show[{NullCurveB, GaugeB}]
```



- **ABABAB··· version of Parrondo's game**

The result of flipping first the A-coin, then the B-coin is described by

```
ℂ = 𝔹.𝔸;
ℂ // MatrixForm
```

$$\begin{pmatrix} x(1-y)+(1-x)z & xz & (1-x)(1-y) \\ (1-x)(1-z) & (1-x)y+x(1-z) & xy \\ xy & (1-x)(1-y) & x(1-y)+(1-x)y \end{pmatrix}$$

NOTE: Products/powers of Markov matrices with 0s on the diagonal typically have—as here—non-zero diagonal elements, which signify not "stand in place" but "return to place."

We look first to construction of the asymptotic stochastic state vector

```
Eigenvalues[ℂ]⟦1⟧
```

1

```
ξ = Simplify[Eigenvectors[ℂ]⟦1⟧]
```

$$\Big\{\big((-1+y)^2+x^2\,(1+y^2+2y(-1+z)-z)-x(-1+y)(-2+2y+z)\big)\Big/\big(x^2\,(1+y^2+2y(-1+z)-z)+(-1+y)(-1+z)-x(-1+y)(-2+y+2z)\big),$$
$$\big(x^2\,(1+y^2+2y(-1+z)-z)+(-1+y)(-1+z)-x(-2+3y)(-1+z)\big)\Big/\big(x^2\,(1+y^2+2y(-1+z)-z)+(-1+y)(-1+z)-x(-1+y)(-2+y+2z)\big),\;1\Big\}$$

```
Denominator[ξ[[1]]] == Denominator[ξ[[2]]]
```

True

```
ξ2 = Simplify[Denominator[ξ[[1]]] ξ]
```

$$\{(-1+y)^2+x^2\left(1+y^2+2y(-1+z)-z\right)-x(-1+y)(-2+2y+z),$$
$$x^2\left(1+y^2+2y(-1+z)-z\right)+(-1+y)(-1+z)-x(-2+3y)(-1+z),$$
$$x^2\left(1+y^2+2y(-1+z)-z\right)+(-1+y)(-1+z)-x(-1+y)(-2+y+2z)\}$$

```
ξ3 = Simplify[Total[ξ2]]
```

$$3x^2\left(1+y^2+2y(-1+z)-z\right)+(-1+y)(-3+y+2z)+x\left(-6-3y^2+y(10-6z)+5z\right)$$

and so are led to the vector

```
PC = Transpose[{ξ2/ξ3}];
```

```
PC // MatrixForm
```

$$\begin{pmatrix} \frac{(-1+y)^2+x^2\left(1+y^2+2y(-1+z)-z\right)-x(-1+y)(-2+2y+z)}{3x^2\left(1+y^2+2y(-1+z)-z\right)+(-1+y)(-3+y+2z)+x\left(-6-3y^2+y(10-6z)+5z\right)} \\ \frac{x^2\left(1+y^2+2y(-1+z)-z\right)+(-1+y)(-1+z)-x(-2+3y)(-1+z)}{3x^2\left(1+y^2+2y(-1+z)-z\right)+(-1+y)(-3+y+2z)+x\left(-6-3y^2+y(10-6z)+5z\right)} \\ \frac{x^2\left(1+y^2+2y(-1+z)-z\right)+(-1+y)(-1+z)-x(-1+y)(-2+y+2z)}{3x^2\left(1+y^2+2y(-1+z)-z\right)+(-1+y)(-3+y+2z)+x\left(-6-3y^2+y(10-6z)+5z\right)} \end{pmatrix}$$

which, as we demonstrate, does possess the required properties:

```
Total[Transpose[PC][[1]]] // Simplify
Simplify[ℂ.PC == PC]
```

1

True

We look next to construction of the components of the asymptotic edge currents—previously defined

$$\begin{pmatrix} \mathbb{M}[[3]][[2]]\,P[[2]][[1]]-\mathbb{M}[[2]][[3]]\,P[[3]][[1]] \\ \mathbb{M}[[1]][[3]]\,P[[3]][[1]]-\mathbb{M}[[3]][[1]]\,P[[1]][[1]] \\ \mathbb{M}[[2]][[1]]\,P[[1]][[1]]-\mathbb{M}[[1]][[2]]\,P[[2]][[1]] \end{pmatrix}$$

—and obtain expressions

```
JC1 = Simplify[ℂ[[3]][[2]] PC[[2]][[1]] - ℂ[[2]][[3]] PC[[3]][[1]]];
JC2 = Simplify[ℂ[[1]][[3]] PC[[3]][[1]] - ℂ[[3]][[1]] PC[[1]][[1]]];
JC3 = Simplify[ℂ[[2]][[1]] PC[[1]][[1]] - ℂ[[1]][[2]] PC[[2]][[1]]];
```

which we verify are identical

```
JC1 == JC2 == JC3
```

True

We are led thus to the construction

```
JC = JC1
```

$$-\left(x^3\left(1+y^2+2y(-1+z)-z\right)+(-1+y)^2(-1+z)-3x(-1+y)^2(-1+z)+3x^2(-1+y)^2(-1+z)\right)\Big/$$
$$\left(3x^2\left(1+y^2+2y(-1+z)-z\right)+(-1+y)(-3+y+2z)+x\left(-6-3y^2+y(10-6z)+5z\right)\right)$$

This gives

```
JC /. {x → 0, y → 0, z → 0}
JC /. {x → 1, y → 1, z → 1}
```

$\frac{1}{3}$

$-\frac{1}{3}$

Note the sign reversal, which I emphasize by comparing

```
JA /. x → 0
JB /. {y → 0, z → 0}
JC /. {x → 0, y → 0, z → 0}
```

$-\frac{1}{3}$

$-\frac{1}{3}$

$\frac{1}{3}$

The following figure shows the JC null surface. Blue/red spheres identify the regions in which JC is positive/negative.

```
NullSurfaceC = ContourPlot3D[JC == 0, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}];
GaugeC = Graphics3D[{
    Line[{{0, 0, 0}, {1, 1, 1}}],
    {Blue, Sphere[{0.25, 0.25, 0.25}, 0.02]},
    {Yellow, Sphere[{0.50, 0.50, 0.50}, 0.02]},
    {Red, Sphere[{0.75, 0.75, 0.75}, 0.02]}}];
Show[{NullSurfaceC, GaugeC}]
```



- **The parameter region that supports Parrondo's Paradox**

We owe to Parrondo the discovery that there exist {x, y, z} parameters for which the A-game and B-game are losing games, but the AB-game is a winning game, which in terms of the present analysis means

```
JA[x, y, z] < 0
JB[x, y, z] < 0
JC[x, y, z] > 0
```

Parrondo himself—go to

```
Hyperlink["Parrondo Home Page", "http://seneca.fis.ucm.es/parr/"]
```

Parrondo Home Page

and open Paradoxical Games > The Seminal Document on the Games: "How to cheat a bad mathematician"—suggests parameters

$x = \frac{1}{2} - 0.005$

$y = \frac{3}{4} - 0.005$

$z = \frac{1}{10} - 0.005$

that do the job

$$\text{JA /. } \left\{x \to \frac{1}{2} - 0.005\right\}$$

$$\text{JB /. } \left\{y \to \frac{3}{4} - 0.005,\ z \to \frac{1}{10} - 0.005\right\}$$

$$\text{JC /. } \left\{x \to \frac{1}{2} - 0.005,\ y \to \frac{3}{4} - 0.005,\ z \to \frac{1}{10} - 0.005\right\}$$

```
-0.00333333

-0.00289843

0.00224585
```

but convey the impression that the "Domain of Paradox" is quite restricted. In this connection, see also Gregory P. Harmer, Derek Abbott, Peter G. Taylor & Juan M. R. Parrondo, "Brownian ratchets and Parrondo's games," Chaos **11**, 705-714 (2001) : Figure 4, page 709, which conveys the same impression.
The present line of argument indicates that the Parrondo phenomenon is actually quite robust, the domain of paradox quite extensive. It can be explored by means of commands

```
{JA, JB, JC};
Manipulate[%, {x, 0, 0.5}, {y, 0, 1}, {z, 0, 1}]
```



in which I have exploited the fact that A-game losing $\Longleftrightarrow 0 < x < \frac{1}{2}$. Look particularly to the wonderfully symmetric paradoxical case $x = y = z = 0.2$, to which I was led by such exploration:

```
JA /. {x → 0.2}
JB /. {y → 0.2, z → 0.2}
JC /. {x → 0.2, y → 0.2, z → 0.2}

-0.2

-0.2

0.2
```

The situation is made vividly clear by the following figure, which shows the JB and JC null surfaces in the half of the unit cube in which JA < 0:

```
SemiBNull = ContourPlot3D[JB == 0, {x, 0, 0.5}, {y, 0, 1}, {z, 0, 1}];
SemiCNull = ContourPlot3D[JC == 0, {x, 0, 0.5}, {y, 0, 1}, {z, 0, 1}];
SymmetricParadoxicalPoint =
  Graphics3D[{{Gray, Sphere[{0, 0, 0}, 0.02]}, {Yellow, Sphere[{0.2, 0.2, 0.2}, 0.02]}}];
Show[{SemiBNull, SemiCNull, SymmetricParadoxicalPoint}, AspectRatio → Automatic]
```



The Paradoxidal Domain is the domain that contains the origin (gray sphere), which in particular contains the "symmetrical point" {0.2, 0.2, 0.2} (yellow sphere). Evidently 25% of the points in the unit cube are paradoxical points.